
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 7.1.2.2.4 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2018-12_D19wk23’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.2.4
Test Group:
\ENDC\7_1
ATS Version:
iwd-TTCN3-B2018-12_D19wk23
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
Qualcomm SDM855+SDX50
Verification Status:
PASS

4. Corrections required for ENDC TC 7.1.2.2.4
Note: Please note that the changes in TTCN CR R5s190317 also apply for this submission. Additional changes are listed here.

Change 4.1
	Function name
	f_TC_7_1_2_2_3and4

	Reason for change
	Step 6 to 10 are repeated for 4095 times and the test case would fail if the advance timing for sending out the DL is not sufficient even for one iteration

	Summary of change
	Increased the advance configuration of DL data transmission to 400 ms to make the test case stable. 

	TTCN module
	RLC_TC_Common_NR

	MCC160 Comment
	


Before Change

  function f_TC_7_1_2_2_3and4(NR_RLC_SS_State_Type p_RLC_Rec,

                              DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { // UM RLC / 6 & 12-bit SN / Correct use of sequence numbering

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var integer v_RLC_SDUsize := 10; // SDU size of 10 octets  @sic R5-185100, R5-193861 sic@

    var integer v_RLC_Len1stSeg := 5; // Length of 1st segment @sic R5-193861 sic@

    var integer v_RLC_UL_Len1stSeg := 5; // Length of 2nd segment for UL

    var integer v_Modulus_UM_SN := f_Power(2, p_RLC_Rec.UM_SN_Size);

    var integer v_RLC_SDUmax := v_Modulus_UM_SN + 1; // Maximum SDU number to be generated

    var integer i;

    var integer v_VerdictFilter; // @sic R5-185100 sic@

    if (p_RLC_Rec.UM_SN_Size == 12) {

      // TBS for the first segment = RLC PDU data + RLC header without SO(2 octets) + MAC header(2 octets)  + Short BSR (2 octets) = 5 + 2 + 2 + 2 = 11 octets => TBS = 88 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO(4 octets) + MAC header(2 octets) = 5 + 4 + 2 = 11 octets => TBS = 88 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);

      v_VerdictFilter := 256;

    } else { //SN6

      // TBS for the first segment = RLC PDU data + RLC header without SO(1 octet) + MAC header(2 octets) + Short BSR (2 octets) = 5 + 1 + 2 + 2 = 10 octets => TBS = 80 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO(3 octets) + MAC header(2 octets) = 5 + 3 + 2 = 10 octets => TBS = 80 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(80, v_NR_UplinkBWP);

      v_VerdictFilter := 16;

    }

    fl_NR_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLC_SDUsize, v_RLC_SDUmax); // v_RLC_SDUmax RLC SDUs

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20); // Timing of send the second segment

    // @siclog "Step 0" siclog@ - prose change needed !

    // The SS stops UL grant transmission

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 1" siclog@

    // The SS transmits UMD PDU#1 with 6/12 bit SN = 0 containing the first segment of RLC SDU#1 (SI field = 01).

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU1, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_TX_Next == 0

    // @siclog "Step 2" siclog@

    // The SS transmits UMD PDU#2 with 6/12 bit SN=0 containing the last segment of RLC SDU#1 (SI field = 10) and including SO field

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU1, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

    // @siclog "Step 3" siclog@

    // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#1 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmit UMD PDU#1 with 6/12 bit SN = 0 containing the first segment of RLC SDU#1 (SI field = 01)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_RX_Next == 0

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // Check: Does the UE transmit UMD PDU#2 with 6/12 bit SN = 0 containing the last segment of RLC SDU#1 (SI field = 10)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    // EXCEPTION:   Steps 6 to 10 are executed 4095 times, the initial value of k = 1, it is incremented by one for each iteration.

    // @siclog "Step 6" siclog@

    // The SS transmits UMD PDU#(2*k+1) with 6/12 bit SN = k containing the first segment of RLC SDU#(k+1) (SI field = 01).

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    for (i := 1; i < v_Modulus_UM_SN; i:= i + 1) {

      // timing to be organised differently, e.g.  based on previous reception? but the timing is not available. would have to be extracted ...

      f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), i, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

      // @siclog "Step 7" siclog@

      // The SS transmits UMD PDU#(2*(k+1)) with 6/12 bit SN=k containing the last segment of RLC SDU#(k+1) (SI field = 10)

      f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), i, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

      f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

      // @siclog "Step 8" siclog@

      // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#(k+1) in 2 RLC/MAC PDUs.

      v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

      // @siclog "Step 9" siclog@

      // Check: Does the UE transmit UMD PDU#(2*k+1) with 6/12 bit SN = k containing the first segment of RLC SDU#(k+1) (SI field = 01)?

      f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, i, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU);

      if (((i+1) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9" & " i = " & int2str(i))}; // @sic R5-185100, R5-187105, R5s190293 sic@

      // @siclog "Step 10" siclog@

      // Check: Does the UE transmit UMD PDU#(2*(k+1)) with 6/12 bit SN = k containing the last segment of RLC SDU#(k+1) (SI field = 10) and including SO field?

      f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, i, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

      f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

      if (((i+1) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10" & " i = " & int2str(i))}; // @sic R5-185100, R5-187105, R5s190293 sic@

      // timing for next transmissions

      v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); //@sic R5s190292 sic@
      v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    }

    // @siclog "Step 11" siclog@

    // The SS transmits UMD PDU#129/8193 with 6/12 bit SN = 0 containing the first segment of RLC SDU#64/4097 (SI field = 01).

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    // timing to be organised differently ? based on previous reception ?

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_Modulus_UM_SN, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);   //@sic R5s190292 sic@

    // @siclog "Step 12" siclog@

    // The SS transmits UMD PDU#130/8194 with 6/12 bit SN=0 containing the last segment of RLC SDU#65/4097 (SI field = 10) and including SO field

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), v_Modulus_UM_SN, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);   //@sic R5s190292 sic@

    f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

    // @siclog "Step 13" siclog@

    // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#65/4097 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20); //@sic R5s190321 sic@

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 14" siclog@

    // Check: Does the UE transmit UMD PDU#8193 with 6/12 bit SN = 0 containing the first segment of RLC SDU#65/4097 (SI field = 01)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, v_Modulus_UM_SN, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU);  //@sic R5s190292 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15" siclog@

    // Check: Does the UE transmit UMD PDU#8194 with 6/12 bit SN = 0 containing the last segment of RLC SDU#65/4097 (SI field = 10) and including SO field?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, v_Modulus_UM_SN, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true); //@sic R5s190292 sic@

    f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

  }


After Change
     function f_TC_7_1_2_2_3and4(NR_RLC_SS_State_Type p_RLC_Rec,

                              DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { // UM RLC / 6 & 12-bit SN / Correct use of sequence numbering

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var integer v_RLC_SDUsize := 10; // SDU size of 10 octets  @sic R5-185100, R5-193861 sic@

    var integer v_RLC_Len1stSeg := 5; // Length of 1st segment @sic R5-193861 sic@

    var integer v_RLC_UL_Len1stSeg := 5; // Length of 2nd segment for UL

    var integer v_Modulus_UM_SN := f_Power(2, p_RLC_Rec.UM_SN_Size);

    var integer v_RLC_SDUmax := v_Modulus_UM_SN + 1; // Maximum SDU number to be generated

    var integer i;

    var integer v_VerdictFilter; // @sic R5-185100 sic@

    if (p_RLC_Rec.UM_SN_Size == 12) {

      // TBS for the first segment = RLC PDU data + RLC header without SO(2 octets) + MAC header(2 octets)  + Short BSR (2 octets) = 5 + 2 + 2 + 2 = 11 octets => TBS = 88 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO(4 octets) + MAC header(2 octets) = 5 + 4 + 2 = 11 octets => TBS = 88 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);

      v_VerdictFilter := 256;

    } else { //SN6

      // TBS for the first segment = RLC PDU data + RLC header without SO(1 octet) + MAC header(2 octets) + Short BSR (2 octets) = 5 + 1 + 2 + 2 = 10 octets => TBS = 80 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO(3 octets) + MAC header(2 octets) = 5 + 3 + 2 = 10 octets => TBS = 80 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(80, v_NR_UplinkBWP);

      v_VerdictFilter := 16;

    }

    fl_NR_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLC_SDUsize, v_RLC_SDUmax); // v_RLC_SDUmax RLC SDUs

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 400); // Timing of send the first segment 
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20); // Timing of send the second segment

    // @siclog "Step 0" siclog@ - prose change needed !

    // The SS stops UL grant transmission

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 1" siclog@

    // The SS transmits UMD PDU#1 with 6/12 bit SN = 0 containing the first segment of RLC SDU#1 (SI field = 01).

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU1, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_TX_Next == 0

    // @siclog "Step 2" siclog@

    // The SS transmits UMD PDU#2 with 6/12 bit SN=0 containing the last segment of RLC SDU#1 (SI field = 10) and including SO field

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU1, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

    // @siclog "Step 3" siclog@

    // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#1 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 4" siclog@

    // Check: Does the UE transmit UMD PDU#1 with 6/12 bit SN = 0 containing the first segment of RLC SDU#1 (SI field = 01)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_RX_Next == 0

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@

    // Check: Does the UE transmit UMD PDU#2 with 6/12 bit SN = 0 containing the last segment of RLC SDU#1 (SI field = 10)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    // EXCEPTION:   Steps 6 to 10 are executed 4095 times, the initial value of k = 1, it is incremented by one for each iteration.

    // @siclog "Step 6" siclog@

    // The SS transmits UMD PDU#(2*k+1) with 6/12 bit SN = k containing the first segment of RLC SDU#(k+1) (SI field = 01).

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 400); 
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    for (i := 1; i < v_Modulus_UM_SN; i:= i + 1) {

      // timing to be organised differently, e.g.  based on previous reception? but the timing is not available. would have to be extracted ...

      f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), i, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

      // @siclog "Step 7" siclog@

      // The SS transmits UMD PDU#(2*(k+1)) with 6/12 bit SN=k containing the last segment of RLC SDU#(k+1) (SI field = 10)

      f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), i, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

      f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

      // @siclog "Step 8" siclog@

      // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#(k+1) in 2 RLC/MAC PDUs.

      v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

      // @siclog "Step 9" siclog@

      // Check: Does the UE transmit UMD PDU#(2*k+1) with 6/12 bit SN = k containing the first segment of RLC SDU#(k+1) (SI field = 01)?

      f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, i, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU);

      if (((i+1) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9" & " i = " & int2str(i))}; // @sic R5-185100, R5-187105, R5s190293 sic@

      // @siclog "Step 10" siclog@

      // Check: Does the UE transmit UMD PDU#(2*(k+1)) with 6/12 bit SN = k containing the last segment of RLC SDU#(k+1) (SI field = 10) and including SO field?

      f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, i, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

      f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

      if (((i+1) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10" & " i = " & int2str(i))}; // @sic R5-185100, R5-187105, R5s190293 sic@

      // timing for next transmissions

      v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 400); //@sic R5s190292 sic@ 
      v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    }

    // @siclog "Step 11" siclog@

    // The SS transmits UMD PDU#129/8193 with 6/12 bit SN = 0 containing the first segment of RLC SDU#64/4097 (SI field = 01).

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 400);
    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    // timing to be organised differently ? based on previous reception ?

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_Modulus_UM_SN, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);   //@sic R5s190292 sic@

    // @siclog "Step 12" siclog@

    // The SS transmits UMD PDU#130/8194 with 6/12 bit SN=0 containing the last segment of RLC SDU#65/4097 (SI field = 10) and including SO field

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), v_Modulus_UM_SN, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);   //@sic R5s190292 sic@

    f_NR_RLC_SS_State_UM_TX_Next_Incr(p_RLC_Rec); // UM_TX_Next incrementation

    // @siclog "Step 13" siclog@

    // SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#65/4097 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20); //@sic R5s190321 sic@

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 14" siclog@

    // Check: Does the UE transmit UMD PDU#8193 with 6/12 bit SN = 0 containing the first segment of RLC SDU#65/4097 (SI field = 01)?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, v_Modulus_UM_SN, 0, v_RLC_UL_Len1stSeg, tsc_NR_SI_StartOfSDU);  //@sic R5s190292 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15" siclog@

    // Check: Does the UE transmit UMD PDU#8194 with 6/12 bit SN = 0 containing the last segment of RLC SDU#65/4097 (SI field = 10) and including SO field?

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, v_Modulus_UM_SN, v_RLC_UL_Len1stSeg, v_RLC_SDUsize - v_RLC_UL_Len1stSeg, tsc_NR_SI_EndOfSDU, true); //@sic R5s190292 sic@

    f_NR_RLC_SS_State_UM_RX_Next_Incr(p_RLC_Rec); // UM_RX_Next incrementation

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

  }
Change 4.2
	Function name
	TC_7_1_2_2_4_ENDC

	Reason for change
	The guard timer is not enough to complete the test case. 

	Summary of change
	Increased the guard timer to 2500 s

	TTCN module
	ENDC_Testsuite

	MCC160 Comment
	


Before Change
  testcase TC_7_1_2_2_4_ENDC() runs on MTC_ENDC system SYSTEM_ENDC {

    // @purpose

    //   UM RLC / 12-bit SN / Correct use of sequence numbering

    var ENDC_NR_PTC      v_ENDC_NR    := null;

    var EUTRA_5GS_PTC    v_EUTRA_5GS  := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(2000);
    v_ENDC_NR := ENDC_NR_PTC.create alive;

    v_EUTRA_5GS := EUTRA_5GS_PTC.create alive;

    f_MTC_ConnectPTCs_ENDC(system, v_ENDC_NR, v_EUTRA_5GS, v_IMS1, v_IMS2);

    v_ENDC_NR.start(f_TC_7_1_2_2_4_ENDC_NR());

    v_EUTRA_5GS.start(f_TC_7_1_2_2_4_ENDC_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop_ENDC(t_GuardTimer);

  }
After Change
  testcase TC_7_1_2_2_4_ENDC() runs on MTC_ENDC system SYSTEM_ENDC {

    // @purpose

    //   UM RLC / 12-bit SN / Correct use of sequence numbering

    var ENDC_NR_PTC      v_ENDC_NR    := null;

    var EUTRA_5GS_PTC    v_EUTRA_5GS  := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(2500);
    v_ENDC_NR := ENDC_NR_PTC.create alive;

    v_EUTRA_5GS := EUTRA_5GS_PTC.create alive;

    f_MTC_ConnectPTCs_ENDC(system, v_ENDC_NR, v_EUTRA_5GS, v_IMS1, v_IMS2);

    v_ENDC_NR.start(f_TC_7_1_2_2_4_ENDC_NR());

    v_EUTRA_5GS.start(f_TC_7_1_2_2_4_ENDC_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop_ENDC(t_GuardTimer);

  }
5. Execution Log Files

5.1 Qualcomm SDM855+SDX50
The Qualcomm SDM855+SDX50 UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on EN-DC band combination DC_5A_n261A.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_7_1_2_2_4_Log.html
· 

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

